# Copyright(C) 2023  Sutou Kouhei <kou@clear-code.com>
#
# This library is free software; you can redistribute it and/or
# modify it under the terms of the GNU Lesser General Public
# License as published by the Free Software Foundation; either
# version 2.1 of the License, or (at your option) any later version.
#
# This library is distributed in the hope that it will be useful,
# but WITHOUT ANY WARRANTY; without even the implied warranty of
# MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the GNU
# Lesser General Public License for more details.
#
# You should have received a copy of the GNU Lesser General Public
# License along with this library; if not, write to the Free Software
# Foundation, Inc., 51 Franklin Street, Fifth Floor, Boston, MA  02110-1301  USA

@PACKAGE_INIT@

set(GRN_WITH_MESSAGE_PACK "@GRN_WITH_MESSAGE_PACK@")
set(GRN_WITH_BUNDLED_MESSAGE_PACK "@GRN_WITH_BUNDLED_MESSAGE_PACK@")
set(GRN_PLUGINS_DIR "@GRN_PLUGINS_DIR@")

include(CMakeFindDependencyMacro)

find_dependency(Threads)
if(DEFINED CMAKE_MODULE_PATH)
  set(GRN_CMAKE_MODULE_PATH_OLD ${CMAKE_MODULE_PATH})
else()
  unset(GRN_CMAKE_MODULE_PATH_OLD)
endif()
set(CMAKE_MODULE_PATH "${CMAKE_CURRENT_LIST_DIR}")
if(GRN_WITH_MESSAGE_PACK AND NOT GRN_WITH_BUNDLED_MESSAGE_PACK)
  find_dependency(Groongamsgpackc)
endif()
if(DEFINED GRN_CMAKE_MODULE_PATH_OLD)
  set(CMAKE_MODULE_PATH ${GRN_CMAKE_MODULE_PATH_OLD})
  unset(GRN_CMAKE_MODULE_PATH_OLD)
else()
  unset(CMAKE_MODULE_PATH)
endif()

include("${CMAKE_CURRENT_LIST_DIR}/GroongaTargets.cmake")

check_required_components(Groonga)
